

Writing Maple-graded questions and questions with Maple plots

Maple T.A. can use the Maple engine to grade students' responses. One advantage of this is that the system will give credit for any response that is mathematically equivalent to the correct answer. For example, if your Maple-graded question were

$$\frac{d}{dx} \sec(x)$$

then the system would accept $\sec(x)\tan(x)$, $\tan(x)/\cos(x)$, $\sin(x)/\cos^2(x)$ or any equivalent as correct. If your question asked for a set, the system would accept any response that contained the correct elements, regardless of the order in which the student gave them. You can write questions that ask for differential equations, unevaluated integrals, and so on. You can write questions that have more than one answer or infinitely many answers.


This document will guide you through the creation of some example Maple-graded questions. The examples reflect only a small sampling of the possibilities.

Example 1: A basic Maple-graded question

In this first example, we build a question that asks for the derivative of $\sec(x)$ and uses Maple to grade the response.

1. Click **Add a new question** in the left half of the Question Bank Editor screen.
2. Choose "Maple-graded" from the **Question Type** menu and type a description in the **Question Description** text box.
3. If you want to include math expressions or plots in the question text, select the check box next to **Use HTML formatting for the question text**.

The screen now looks like this.



4. Click the **Next** button. This will take you to a screen in which you type the question text and the Maple code for grading the student response.
5. Type the question in the first text area, for example, “Compute the derivative of $\sec(x)$.” (To write this as a formatted mathematical expression, see the section ***Including formatted math expressions in a question*** below.)
6. Type one or more lines of Maple code in the second text area that will grade the student response. In this example, a single line of code is sufficient:

evalb(simplify(\$RESPONSE - diff(sec(x),x)) = 0);


The Maple engine executes the lines of code in order when the student submits the response. The system stores the student response in the variable \$RESPONSE, which you can use as a Maple variable anywhere inside your Maple code.

The last line of your Maple code must evaluate to **true** (for a correct response) or **false** (for an incorrect response). Maple’s **evalb** command is frequently the best way to do this, but any Maple command that returns a boolean is allowed.

Caution: We strongly recommend that you test your Maple code in a Maple worksheet. Frequently you will find that your first version of Maple code is inadequate to give credit for all correct answers. In this example, if we had just used **evalb(\$RESPONSE = diff(sec(x),x))**, the system would only give credit for $\sec(x)\tan(x)$ and reject all other responses, even if algebraically equivalent to $\sec(x)\tan(x)$. The reason is that **evalb** does not perform any simplification.


7. Select **Formula - e.g. $x^2\sin(x^2)$** from the menu under the Maple code box. This option lets students enter basic algebraic expressions, either by typing them in standard calculator syntax, e.g. $(2x/y + \sin(xy))(4x^2-1)$, or using a symbol palette. When the student submits the response in this form, the system translates it into Maple syntax so that the Maple engine can apply your Maple code to it.

If you want the student to enter equations, differential operators, integral signs, summations, vectors, matrices, DE’s, or other advanced mathematical expressions, then select the option **Maple syntax – e.g. $\text{diff}(2*f(x),x)$** . This option allows a much wider range of student response types. However, it requires the student to enter the response using proper Maple syntax, because the system runs your Maple code directly on what the student types.



8. Click the **Finish** button. This will take you to a screen in which you can test your question or start a new one.
9. Now pretend you are the student doing the assignment and enter a typical response in the box.

The screen would look like this.



10. Click the **Grade** button to check that Maple is correctly assessing the response. Click the **Ok** button to take you back to the previous screen. Try a variety of student responses to convince yourself that the

Maple code for grading the question is adequate. If the question or the Maple code needs revision, click the **Edit** button.


Example 2: A randomized question

To provide multiple review problems of the same form or to hinder cheating during exams, you can put random parameters such as random problem data or coefficients into math questions.

In this example, we build a question that asks for the equation of a 3-D plane that passes through the points $\langle 1, 0, 0 \rangle$, $\langle 0, 1, 0 \rangle$ and a third point picked at random.

This example will also show how to use multi-line Maple code to grade a more complicated question.

1. Click **Add a new question**, pick Maple-graded and type a description of the problem.
2. Click the **Add** button next to the heading **This algorithm will generate multiple versions of the question**. This will take you to a screen in which you define the random variables.
3. Click the **Show Designer** button. This will take you to a point-and-click interface for designing random variables. (Once you become familiar with Maple T.A.'s syntax for designing random variables, you can simply type the randomization code into the text area and skip the **Show Designer** step.)
4. Here we will define three random coordinates for the third point. In this example, we call them **a**, **b** and **c**. Enter the name of the variable in the first text box. Select the data type (usually whole number) from the adjacent menu. Enter the lower and upper bounds of the variable in the next two text boxes. Click the [OK](#) label to the right. Do this for **a**, **b** and **c**. The screen will look like this.



5. When you are finished defining variables, click the **Ok** button at the top of the screen.
6. Click **Next** in the following screen to get to the question text area.
7. In the question text area and in the Maple code that grades the response, refer to the random variables using the names \$a, \$b and \$c. For this example, the question text would be:


Find the equation of the plane that passes through $\langle 1,0,0 \rangle$, $\langle 0,1,0 \rangle$ and $\langle \$a, \$b, \$c \rangle$. Write your answer in the form $Ax+By+Cz+D=0$ (using lower case x, y and z).

The Maple code for grading the student response would be:


```
n := LinearAlgebra:-CrossProduct(<$a,$b,$c>-<1,0,0>, <0,1,0>-<1,0,0>);
plane := n[1]*(x-1)+n[2]*y+n[3]*z=0;
evalb( lhs($RESPONSE)=lhs(plane) or lhs($RESPONSE)=-1*(lhs(plane)) );
```

8. Set the type of expression students will enter as **Maple syntax** – e.g. `diff(2*f(x),x)`, because students will be entering an equation. (**Caution:** If this option is set to **Formula** e.g. `x^2*sin(x)`, the system will not parse the equal-sign in the equation the student enters and will grade the response as incorrect.)

The screen now looks like this.



9. Click the **Finish** button. When the question appears, notice that the third point has been generated randomly by Maple T.A.



Random Maple variables

You can also create random Maple variables. Enter them using the Designer interface (in the third row of text areas), or by typing them directly in the question's source code as

`$<var name> = maple("<maple command>");`


For example, to create a random polynomial called `p` using the Designer, enter `p` as the variable name and the following as the Maple command in the third row of the Designer window:

`randomize():RandomTools:-Generate(polynom(rational, x, degree=3))`

No semicolon is necessary inside the text box. Click the link [OK](#) in that row. You can then refer to the polynomial as `$p` in the Maple code, the question text, and subsequently defined random variables.

(The prefix "`randomize():`" is required for a different instance of the polynomial to be generated upon each instance of the question. Without it, the same Maple object would be generated each time.)

The screen would look like this:



Caution: It is often necessary to wrap a random variable inside parentheses when referring to it inside a Maple statement. For example, if you have a random variable \$a, and your Maple code contains the lines,

```
k := 2*$a + x^$a;
evalb( k < 10 );
```

Maple will generate a syntax error if \$a turns out to be negative, for Maple cannot parse $2 \cdot -3$ or x^{-3} . *Maple T.A. does not notify you of Maple-generated errors.* It will simply mark every response wrong because the **evalb** statement will never evaluate to **true**.

To avoid this, revise the first line of your Maple code to

```
k := 2*($a) + x^($a);
```

We recommend wrapping all random variables inside parentheses when referring to them inside Maple code.

More examples of Maple-graded questions

Example 3: A question with infinitely many correct answers

Question text:

Give a function $f(x)$ that is increasing on the interval $[0,10]$.

Maple code:

```
assume(x >=0, x <= 10);  
verify(diff($RESPONSE, x), 0, {'greater_equal'});
```

Type of expression to accept:

Formula.

This question has infinitely many correct answers. We can use Maple's **verify** command to check that the derivative of the student's response is greater than or equal to 0 for all values of x under our assumption.

Example 4: Another question with infinitely many correct answers

Question text:

Use integration by parts to give an antiderivative of the function $f(x) = x \sin(x)$.

Maple code:

```
evalb( diff($RESPONSE, x) = x*sin(x) );
```

Type of expression to accept:

Formula.

The system will mark correct any response of the form $\sin(x) - x \cos(x) + C$, where C is a constant, because we are asking Maple to differentiate the response. (The code `evalb(simplify($RESPONSE - int(x*sin(x),x)) = 0)` is not appropriate here, for it would mark *incorrect* any response in which $C \neq 0$.)

We use **Formula** as the type of expression to accept, in order to prevent Maple-savvy students from entering “int($x \sin(x)$, x)” as the response.

Example 5: Accepting approximations of exact symbolic answers

Question text:

Find the perimeter of the circle with radius r correct to four decimal places.

Maple code:

```
exact := 2*Pi*(r);  
approx := evalf( exact, 4);  
evalb( abs( evalf($RESPONSE - approx)/ approx ) < .05 );
```

Type of expression to accept:

Formula.

This problem will mark correct either the exact symbolic expression $2r$ or any numeric response that is within 5% of $2r$ evaluated to four decimal places. Notice that the radius r of the circle is randomly generated. Notice also that we have wrapped r inside parentheses in the definition of **exact**.

Example 6: Questions with vectors and matrices

Question text:

In what direction from the point $\langle p_1, p_2 \rangle$ is the function $f(x,y) = x^2y^2$ increasing the fastest? Express your answer as a vector in the form $\langle a, b \rangle$.

Maple code:

```
gradientField := VectorCalculus:-VectorField(<2*x*y^2, 2*y*x^2>, cartesian[x,y]);  
gradientValue := VectorCalculus:-evalVF( gradientField, <$p1, $p2>);  
LinearAlgebra:-Equal( $RESPONSE, gradientValue );
```

Type of expression to accept:

Maple syntax.

Notice that the last line of Maple code uses the **LinearAlgebra:-Equal** command, which takes two vectors or matrices and returns **true** if and only if they have the same entries. The command **evalb(\$RESPONSE = gradValue)** would not work in this example because the “=” sign in Maple cannot be applied to Matrices or Vectors.

We set the type of expression to accept as **Maple syntax** instead of **Formulas** because the **Formulas** option does not parse vector brackets < >.

Troubleshooting Maple-graded questions

Problem	Solution
“T.A. keeps grading correct responses as incorrect..”	<ol style="list-style-type: none">1. Test your Maple code in a Maple worksheet on several forms of the correct response.2. If you used Maple commands from a Maple package, make sure you included the package name with the commands. For example, instead of evalb(Norm(<\$a,\$b>, 2) = 1); you should use evalb(LinearAlgebra:-Norm(<\$a,\$b>, 2) = 1);3. Check that all random variables in the Maple code have a \$ in front of them.4. Check that all random variables in the Maple code have parentheses around them. For example, instead of p := factor(x+2*\$b); use p := factor(x+2*(\$b));5. Check the type of expression to accept. If the correct answer is (or contains) an equation, vector, matrix, unevaluated derivative or integral, or a Maple command, the type of expression must be set to Maple syntax, not Formula.6. Check that all references to \$RESPONSE are in all upper-case letters.
“I’ve just switched the Math Editor Mode from text mode to symbolic mode, but the input box for the response is still in text mode”	The mode does not automatically update in your question-testing window. To test the question in the new mode, refresh by clicking Edit , then Edit again, then Finish .

“I’m seeing variable names instead of instantiated variable values in the question.”	Make sure that all references to random variables have a \$ character in front of them.
“My random Maple variables have the same values at each question instance.”	Make sure the Maple command that generates the random object has the prefix randomize(): The syntax is \$<var name> = maple(“randomize(): <maple command>”);

Including formatted math expressions in a question

Maple T.A. renders the question text as HTML/MathML if the **Use HTML** check box has been selected. This allows you to include formatted math expressions in questions.

In this section, we enhance Example 1 by making the question text appear as


Compute $\frac{d}{dx} \sec(x)$

instead of


Compute the derivative of sec(x).

The easiest way to include a formatted math expression in a Maple-graded question is to use Maple’s **MathML:-ExportPresentation** command as follows.


1. Navigate to the question text area (Step 5 in the previous section) and type the text that precedes the math expression.
2. Open a Maple worksheet and type the Maple code for the math expression you want to display.
3. Call **MathML:-ExportPresentation** on the Maple output. The Maple worksheet might look like this:



4. Copy the resulting MathML string to the clipboard and paste it into the Maple T.A. question text area.
5. Remove the quotation marks from the MathML string. **This step is important!** The screen now looks like this.



6. Click **Finish**. The question will now display the question with pretty-print mathematics, as shown in the following picture.



Formatted math expressions in other question types

For the “Mathematical formula” question type, Maple T.A. provides a graphical MathML editor.

To use the MathML editor in a Mathematical formula question:

1. Advance to the screen in which you enter the question text.
2. Click the **Edit** button. This brings you to a MathML editor screen.
3. Using the buttons in the right-hand panel, create your expression.
4. When you are finished, move the MathML expression into the left-hand panel by clicking the left-arrow button.
5. Click the **Ok** button.

The editor uses WebEQ technology to translate the graphical form of your expression to MathML code.

You may also use the MathML editor as an alternative to Maple worksheets for inserting MathML into Maple-graded questions, although it requires more steps. You can do this as follows:

1. Create a new question of type Mathematical formula
2. Generate MathML code using the MathML editor as described above.
3. Copy and paste the MathML string into your Maple-graded question.

Note: In a later revision of Maple T.A., the MathML editor will be available from within Maple-graded questions, as well.

Formatted math expressions that contain random variables

If you want an algorithmic variable inside a formatted math expression, you must first create the expression using the plain variable name (e.g. “a”) and then edit the MathML code inside Maple T.A. to prepend the \$ (e.g. “\$a”). This is necessary because neither Maple nor WebEQ accepts variable names that start with \$.

For example, to create the question,

Find the determinant of $\begin{bmatrix} a & c \\ b & d \end{bmatrix}$

where a, b, c and d are randomly generated integers, do the following:

1. Create the random integers \$a, \$b, \$c and \$d inside Maple T.A.
2. Inside a Maple worksheet, generate the MathML code for the matrix using **MathML:-ExportPresentation(<<a,b>|<c,d>>)** and paste the string into the question text box.
3. Inside the question text box, insert a \$ in front of each of a, b, c and d in the MathML string.
4. Remove the quotation marks from around the MathML string.

Sometimes the ideal form of the math expression depends on the random variable's value. For example, if the random integer \$n turns out to be 0, then $2 \cdot x^{(n)}$ would ideally be displayed to the student as "2", not " $2x^0$ " (which it would if you followed the above procedure). In such cases, create a separate random variable that calls **MathML:-ExportPresentation** on the random expression.

In this example, you would define the following three random variables in the Designer:

```
$n = range( 0, 3);  
$xn = maple("x^($n)");  
$display_xn = maple(" MathML:-ExportPresentation(2*($xn)) ");
```

Then write your question text in terms of the variable \$display_xn. For example:

What is the degree of \$display_xn?

The polynomial will be displayed with double quotes around it, but it will be properly formatted otherwise.

We recommend this method *only* when you require simplification of the math expression after substitution of the random variables, so as not to add an unnecessary call to **MathML:-ExportPresentation** to the server's load every time the question gets rendered. There is an unavoidable overhead with every call to Maple.


Troubleshooting formatted math expressions

Problem	Solution
"I'm seeing raw MathML tags but no formatted math in the question."	<ol style="list-style-type: none">1. Make sure that the Use HTML box is checked.2. Make sure that the MathML string does not have quotation marks around it.
"I'm seeing variable names instead of instantiated variable values in the question."	Make sure that all references to random variables have a \$ character in front of them.

Inserting Maple plots in questions

You can include Maple plots in any question you create in Maple T.A. As an example, suppose you wish to present the student with the plot of a function $f(x,y)$ and have the student pick the level curves that correspond

to $f(x,y)$ from a set of four choices. The question would appear to the student in multiple-choice format, as shown below. (Only the first choice is visible.)



To insert a plot in any question type:

1. Make sure the **Use HTML** box is checked.
2. Type the instructions to the student in the question text area.
3. In the place in the question text where you want the plot to appear, type
`<script> drawMaplePlot(" Maple-plot-command goes here"; </script>`
4. Enter the code for the correct answer and click **Finish**. You should then see the question with the plot appear.

For the example shown above, the question text would be:

Which of the following show level curves of this graph?

```
<script>drawMaplePlot("plot3d((x^2+2*y^2)*exp(1-x^2-y^2), x=-2..2, y=-2..2, axes=framed,
scaling=constrained, grid=[50,50])");</script>
```

You can use any of Maple's plotting commands, including those in Maple's **plots** and **plottools** packages. To use commands from a Maple package, you must include the package name in the command. For example, to use the **spacecurve** command from the **plots** package, type

```
<script>drawMaplePlot("plots[spacecurve]([cos(t),sin(t),t], t=0..2*Pi)");</script>
```

Use the same syntax to include plots in multiple-choice items.

Troubleshooting plots

Problem	Solution
“I get a broken image where the plot should be.”	Make sure that the Maple plot command includes the package name, if any. For instance, instead of <code><script>drawMaplePlot("spacecurve([t,t,t]...</code> you must use <code><script>drawMaplePlot("plots[spacecurve]([t,t,t]...</code>
“Instead of a plot, I’m only seeing the actual <script> command I typed.”	<ol style="list-style-type: none">1. Make sure the Use HTML box is checked.2. Double-check the syntax of the <script> command. Remember the Maple command must be in quotes, the drawMaplePlot() must end with a semicolon, and the line must end with </script>



Corporate Headquarters
Maplesoft, Waterloo, Canada
t. 519.747.2373 | f. 519.747.5284
800.267.6583 (US & Canada)
info@maplesoft.com

European Office
Maplesoft Europe GmbH, Zug, Switzerland
t. +41 (0)41 763 33 11
f. +41 (0)41 763 33 15
info-europe@maplesoft.com

www.maplesoft.com | www.mapleapps.com